Math 2250
Maple Project 1
August 2002

Due date: September 9, 2002.

References: Edwards-Penney, pages 55–57. Code in `maple` appears in `mapleL1-2002-text.mws` at URL `http://www.math.utah.edu/~korevaar/`. This document: `mapleL1-2002.pdf`.

The Problem. The project is to determine an approximation for the indoor temperature $u(t)$ in an unheated building. The model uses Newton cooling, insulation data and a formula for the ambient outside temperature $A(t)$.

Assumptions and Notation. Let the daily temperature in Salt Lake City vary from a low of $m = 20$ degrees to a high of $M = 50$ degrees with the low and high at 3am and 3pm, respectively. The building is assumed insulated, the average insulation factor being represented by a Newton cooling constant k, usually $0.2 \le k \le 0.5$. The ambient outside temperature is called $A(t)$ and the inside temperature is called $u(t)$, where t is in hours measured from midnight $t = 0$. The number u_0 is the temperature at midnight. Modeling assumes that no heat sources or heat sinks are present inside the building.

Newton Cooling Model. Newton's law of cooling is:

> *The rate of change of the indoor temperature is proportional to the difference between the ambient and indoor temperatures.*

This implies that du/dt is proportional to $A - u$. Let $A(t)$ be the ambient outside temperature and let k denote a constant. A model for the indoor temperature $u(t)$ is given by $du/dt = k(A-u)$, $u(0) = u_0$, which can be rearranged as

$$u'(t) + ku(t) = kA(t), \quad u(0) = u_0. \tag{1}$$

The number k is called the **insulation constant**.

Ambient Temperature Model. Let $M = 50$, $m = 20$, $\omega_0 = \pi/12$. The ambient temperature formula

$$A(t) = \frac{1}{2}(M+m) - \frac{1}{2}(M-m)\cos\omega_0(t-3)$$

satisfies $M = \max A(t) = A(15)$, $m = \min A(t) = A(3)$ and $A(t)$ is t-periodic of period 24 hours.

For use in `maple`, a function of two variables

$$AA(t,\omega) = \frac{1}{2}(M+m) - \frac{1}{2}(M-m)\cos\omega(t-3)$$

is used to maintain the variable name ω in displays.

Indoor Temperature $u(t)$. The integrating factor method for linear equations applies to find the general solution by these steps:

$u' + ku = kA(t)$	Copy the differential equation. The integrating factor is e^{kt}.
$\dfrac{(e^{kt}u)'}{e^{kt}} = kA(t)$	Replace the left side.
$(e^{kt}u)' = kA(t)e^{kt}$	Clear fractions.
$e^{kt}u = u_0 + \int_0^t kA(r)e^{kr}\,dr$	Apply the method of quadrature and use $u(0) = u_0$.
$u = u_0e^{-kt} + e^{-kt}\int_0^t kA(r)e^{kr}\,dr$	Divide to isolate u.
$u = u_0e^{-kt} + k\int_0^t e^{k(r-t)}A(r)\,dr$	Exponential voodoo: $e^ae^b = e^{a+b}$.

Let $u_h(t) = u_0e^{-kt}$, a solution of the homogeneous differential equation $u' + ku = 0$. Let $u_p(t) = k\int_0^t e^{k(x-t)}A(x)dx$, a particular solution of the nonhomogeneous differential equation $u' + ku = kA(t)$. Then the indoor temperature $u(t) = u_h(t) + u_p(t)$ depends on the time t, the initial temperature u_0 and the insulation constant k. Write $u = u(t, u_0, k)$ to emphasize the dependence. In `maple`, advantages exist for adding the variable name ω, which is later set to value $\omega_0 = \pi/12$. Write $u = U(t, u_0, k, \omega)$ for use in `maple`.

Steady-state solution. The steady-state solution u_{SS} is obtained from the general formula $u = u_h + u_p$ by dropping all terms containing a negative exponential. It depends on t and k only.

Problem 1.1. (Solution formulas for u_p and u)

Give an explicit symbolic formula for $u_p(t)$. Display a final formula for $u = u_h + u_p$ which depends only on t, u_0 and k. Check the answer for u in `maple`. The only `maple` assist in this problem is the answer check.

Problem 1.2. (Ambient Temperature Plot)

Plot $A(t)$ on $0 \le t \le 24$ using `maple`. Pencil-in labels for the maxima and minima. Verify by hand that $A(t)$ has period equal to 24.

Problem 1.3. (Indoor Temperature $u(t)$)

Use `maple` to make one graphic with four plots of u versus t, $0 \le t \le 72$ hours. Use $k = 0.3$, and for the four plots midnight temperatures $u_0 = 40$, $u_0 = 50$, $u_0 = 60$, $u_0 = 70$. Hand-label the maxima and minima.

Problem 1.4. (Steady-state Periodic Solution)

Find a formula for the steady-state periodic solution u_{SS} of $u' + ku = kA(t)$. The only `maple` assist in this problem is an answer check.

Problem 1.5. (Indoor-Outdoor Variation)

Compare in a `maple` graphic the indoor and outdoor temperature oscillations over a 48-hour period assuming $k = 0.3$, $u_0 = 69$. Compute the indoor temperature variation from this plot. Find the phase delay using a second plot of steady-state and outdoor temperatures.

Problem 1.6. (Freezing Pipes)

Assume the insulation constant k ranges from 0.2 to 0.5. Suppose the inside temperature is 69 degrees at midnight when the furnace is turned off. Determine approximate ranges of hours during which the indoor temperature is at or below 30 degrees. Illustrate with a computer graphic.

Problem notes.

Notes on 1.1: A hand solution is expected for the integration of u_p. The answer check in `maple` is organized as follows. The complications of setting $\omega = \pi/12$ are avoided here by leaving ω as a symbol, since it does not affect the answer check.

```
# Test LHS=RHS for u'+ku=kA.
M:=50:m:=20:t:='t':omega:='omega':u0:='u0':k:='k':
AA:=(t,omega)->(M+m)/2-(M-m)*(1/2)*cos(omega*(t-3)):
uh:=u0*exp(-k*t):
up:=k*int(exp(k*(x-t))*AA(x,omega),x=0..t):
u:=uh+up:
LHS:=diff(u,t)+k*u:
RHS:=k*AA(t,omega):
simplify(expand(LHS-RHS));
```

A successful test of $LHS = RHS$ produces answer *zero*, or an expression that reduces to zero.

Notes on 1.2: The maximum and minimum occur when the cosine term is smallest and largest, which is when the argument of the cosine is π or 0, respectively. To show that $A(t)$ is periodic of period 24, observe that sums of T-periodic functions are T-periodic, and that $\cos \omega t$ is known from trigonometry to have period $2\pi/\omega$. To plot, use `plot(AA(t,Pi/12),t=0..24);` in `maple`, with `AA` defined as in 1.1 above.

Notes on 1.3: The solution $u = U(t, u_0, k, \omega)$ obtained in 1.1 can be programmed in `maple` as follows:

```
M:=50:m:=20:t:='t':u0:='u0':k:='k':omega:='omega':
AA:=(t,omega)->(M+m)/2-(M-m)*(1/2)*cos(omega*(t-3)):
uh:= u0*exp(-k*t):
up:= k*int(exp(k*(x-t))*AA(x,omega),x=0..t):
U:=unapply(uh+up,(t,u0,k,omega)):
```

For the maple plot, use the following example, which puts two of the four plots onto one graphic.

```
u1:=U(t,40,0.3,Pi/12): u2:=U(t,50,0.3,Pi/12):
plot({u1,u2},t=0..72);
```

Notes on 1.4: The steady-state solution is found from the hand-generated symbolic solution $u = u(t, u_0, k)$ in 1.1 by dropping all terms that contain e^{-kt}. The answer, where $\omega_0 = \pi/12$:

$$u_{SS} = 35 - \frac{15k}{k^2 + \omega_0^2}(k \cos \omega_0(t-3) + \omega_0 \sin \omega_0(t-3)).$$

To *check the answer*, use maple as in 1.1. To get maple to report the above formula, it is essential to evaluate $U(t, u_0, k, \omega)$ and then set ω equal to $\omega_0 = \pi/12$, to disallow expansion of the symbol ω.

Notes on 1.5: The two plots are placed onto one graphic by this maple command:

```
plot({U(t,69,0.3,Pi/12),AA(t,Pi/12)},t=0..48);
```

Click with the left mouse button on the high and low spots in the graphic. Somewhere on the maple worksheet the coordinates of the click are displayed, and this is enough to find a good approximation to the max and min values. The indoor temperature variation is just the maximum minus the minimum.

The phase shift is found from another plot, which uses the steady-state solution $u_{SS}(t)$ obtained in 1.4:

```
SS:=35-(15*k/(k^2+omega^2))*(k*cos(omega*(t-3))+omega*sin(omega*(t-3))):
uss:=unapply(SS,(t,k,omega)):
plot({uss(t,0.3,Pi/12),AA(t,Pi/12)},t=0..48);
```

The shift is computed as $|T_2 - T_1|$, where $A(T_1) = \max A(t)$ and $u_{SS}(T_2) = \max u_{SS}(t)$. Look at the graphic to find sane answers for T_1 and T_2. See the textbook for a more complete discussion of the ideas.

Notes on 1.6: A computer algebra assist for this problem can be found in maple's function implicitplot. This function can plot the equation $u(t, 69, k) = 30$ over the domain $0 \le t \le 72$, $0.2 \le k \le 0.5$. From this plot, and the 3D-plot $z = u(x, 69, y)$, the question is easily answered.

```
with(plots):
M:=50:m:=20:t:='t':u0:='u0':k:='k':omega:='omega':
AA:=(t,omega)->(M+m)/2-(M-m)*(1/2)*cos(omega*(t-3));
Uh:= u0*exp(-k*t):
Up:= k*int(exp(k*(x-t))*AA(x,omega),x=0..t);
U:=unapply(Uh+Up,(t,u0,k,omega)):
implicitplot(U(t,69,k,Pi/12)=30,t=0..72,k=0.2..0.5);
plot3d({U(t,69,k,Pi/12),30},t=0..72,k=0.2..0.5);
```

Zoom in on the plot by using a smaller time domain, suggested by the larger plot. Physically, inside temperature 30 degrees is reached a few hours after the outside temperature drops below 30 degrees. During 72 hours, there are three such inside temperature drops (see the 3D-plot, where z=temperature). For $k < 0.25$, inside temperature 30F is not reached in the first 23 hours.